
Hoofdstuk 1:  Inleiding
1 Tekortkomingen van SQL
Sql:
· 4e generatie taal

· beschrijven wat gedaan moet worden (niet hoe)

2 PL/SQL
= Procedural Language Extensions to SQL
2.1 PL/SQL-uitbreidingen aan SQL
I Variabelen & constanten
· mogelijkheid om variabelen & constanten te declareren
· hieraan waarden toekennen
II Cursors
· bij uitvoeren sql-statement ( werkbuffer (context area) aangemaakt
· controle krijgen over sql-statement & betrekken in programma ( context area een naam geven

· cursor = context area met naam

· mogelijk om resultaattabel van select-statement, rij per rij te doorlopen

III Programmabesturing
· If – then – else
· Loop

· Go – to
2.2 PL/SQL engine
· Nodig om met PL/SQL te werken
· behandeling

· PL/SQL ( procedural statement executor

· SQL ( SQL statement executor
Hoofdstuk 2:  Basisbegrippen
1 PL/SQL – blok
1.1 Componenten (3)
· kan meerdere blokken bevatten

· elk blok afgesloten met sleutelwoord: END; (nieuwe lijn) /
I Declaratie
· optioneel
· steutelwoord: DECLARE
· declareren:
· identifiers ( om gegevens tijdelijk te stockeren 
(bv. variabelen, constanten, cursors & exceptions)
· procedures & functies

· samengestelde datatypes
II Uitvoerbare code

· verplicht
· steutelwoord: BEGIN
· uitvoerbaar commando ( steeds afsluiten met ‘;’

a) Procedure: dbms_output.put_line

dbms_output.put_line (‘[tekst]’)

· om meldingen naar scherm te schrijven

· eerst serveroutput aanzetten

SET SERVEROUTPUT ON
III Exception gedeelte

· optioneel
· steutelwoord: EXCEPTION

· uitvoerbare code bij bepaalde fout at runtime

1.2 Creatie & uitvoering

I Direct ingeven
· afsluiten met END;
II Via editor

a) Rechtstreeks wijzigen bufferinhoud
· oproepen bufferinhoud: EDIT
· inhoud editor overschrijven
b) Sql-bestand aanmaken

· sql-bestand openen: EDIT [bestandsnaam]
· uitvoeren:

· @[bestandsnaam]

· START [bestandsnaam]

· niet opgeslagen als .sql ( bestandsnaam.extensie

1.3 Commentaarregels

· 1 regel of deel: -- [commentaar]
· meerdere regels: /* [commentaar] */
2 Identifiers
a) Constanten
· programmatorisch niet wijzigbare waarden
· enkel lezen
b) Variabelen

· programmatorisch wijzigbare waarden

· lezen & wijzigen
c) Cursors

· uitvoering van select-statements
· resultaattabel rij per rij doorlopen
d) Exceptions
· om runtime fouten op te vangen
· naam bijhouden van code dat uitgevoerd moet worden
e) Procedures, functies & packages

· opgebouwd uit 1 of meerdere PL/SQL blokken
3 Literals

· constante waarden
· gebruikt:

· bij toekenning variabelen & constanten

· in vergelijkingen

a) Numerieke

b) Alfanumerieke

c) Booleaanse

· 3 waarden: true, false, null
4 Expressies
· = uitdrukking waarin identifiers/literals/operatoren met elkaar gecombineerd worden
a) Operatoren
· rekenkundige: +, -, *, /

· alfanumerieke: || (concatenatie)

· vergelijkings:

· relationele: =, <, >, <=, >=, <>, !=, ^=
· bijzondere: is (not) null, like, in, between … and …
· logische: and, or, not
4.1 Gewone expressies
· resultaat: welbepaalde waarde
· gebruikt bij: toekennen van waarden aan constanten & variabelen

4.2 Logische / booleanse expressies

· resultaat: true, false, null
· gebruikt bij:

· if – then – else

· loops

· (niet bij toekenningen)

Hoofdstuk 3:  Constanten & variabelen
1 Declaratie
· om gegevens op te slagen
· eerst declareren in declaratie-gedeelte
1.1 Declaratie van constantan
· steutelwoord: CONSTANT
· verplicht initialisatie (= beginwaarde toekennen)
c_[naam] CONSTANT [datatype] := [expressie];
c_[naam] CONSTANT [datatype] DEFAULT [expressie];
· constante ( initiële waarde mag niet gewijzigd worden

· voorbeeld:

DECLARE

c_maximum CONSTANT number(4) := 1000;

c_minimum CONSTANT number(3) DEFAULT 100;

c_midden CONSTANT number(3) := (c_maximum + c_minimum)/2;

c_tekst CONSTANT varchar2(30) := ‘Het maximum is bereikt!’;

BEGIN

…

END;

1.2 Declaratie van variabelen

· minimaal: naam, datatype
· initialisatie: niet verplicht (anders: NULL)
l_[naam] [datatype] := [expressie];
l_[naam] [datatype] DEFAULT [expressie];
· variabele ( initiële waarde mag gewijzigd worden

· voorbeeld:
DECLARE

l_totaal number(4) DEFAULT 0;

l_teller number(3) := 1;

l_tekst varchar2(30);

BEGIN

…

END;
1.3 Datatypes

a) Numerieke

NUMBER[precisie,schaal]
· getallen van willekeurige grotte

· precisie: totaal # cijfers (max. 38)

· schaal: op hoeveel afgerond

BINARY_INTEGER

· in binair formaat ( rekenkundige bewerkingen verlopen sneller
b) Alfanumerieke

CHAR([lengte])

· vaste lengte

· actuele waarde korter ( rechterkant opgevuld met spaties
· VARCHAR2[lengte]

· variabele lengte

· actuele waarde korter ( waarde exact opgeslagen (geen opvulling met spaties)
c) Datum

DATE
· datum + tijd
d) Logische waarden

BOOLEAN
· waarde: true, false, null
e) Verzameling van gegevens

· zelf datatypes definiëren
· samengestelde datatypes: meerdere gegevens tegelijk opslagen worden
2 Gebruik

· inhoud variabele wijzigen door …, d.m.v. ‘:=’
· literal (= inhoud andere identifier)

· resultaat van expressie

· waarde uit DB d.m.v. select-statement
· conversiefuncties:
· TO_NUMBER([identifier])

· TO_DATE([identifier])

· TO_CHAR([identifier], [format])

Hoofdstuk 4:  Program flow
Programmaverloop bepalen
1 Voorwaardelijke structuur

a) If-then-else

IF [voorwaarde 1] THEN 
…
ELSE 
…
END IF;

b) Elsif

IF [voorwaarde 1] THEN 
…
ELSIF [voorwaarde 2] THEN 
…
ELSE
 … 
END IF;

c) Geneste if-then-else
IF [voorwaarde 1] THEN 
…
ELSE

IF [voorwaarde 2] THEN 
…
ELSE 
… 

END IF;
END IF;
· gebruiker input laten ingeven: &

· voorbeeld:

l_getal := &getal;

…

END;

/

2 Lussen

2.1 Basic loop
LOOP
[te herhalen code]

EXIT WHEN [voorwaarde]

END LOOP;
2.2 While loop

WHILE [voorwaarde] LOOP
[te herhalen code]

END LOOP;
2.3 For loop

· loopindex ( variabele impliciet gedeclareerd in for (bestaat enkel binnen for)

· loopindex automatisch + 1

· REVERSE: loopindex automatisch – 1
· boven- & ondergrens ( 1maal geëvalueerd (bij binnengaan loop)
FOR [loopindex] IN [REVERSE] [ondergrens]..[bovengrens] LOOP
[te herhalen code]

END LOOP;

3 Labels

3.1 GOTO commando
· van bepaald punt, naar ander punt springen (label)
· invoegen label: <<[labelnaam]>>
· springen: GOTO [labelnaam]
3.2 Blok-labels

· vanuit blok refereren naar variabelen uit ander blok
· referentie: [bloknaam].[variabele]
· bloklabel vóór declare invoegen: <<[bloknaam]>>
· afsluiten met: END [bloknaam]

· voorbeeld:

<<hoofdblok>>
DECLARE

…

BEGIN

…

<<subblok>>

DECLARE

l_teller number(2):=1;
BEGIN

…

END subblok;

dbms_output.put_line(subblok.l_teller);

END hoofdblok;

3.3 Loop labels
· looplabel voor begin van loop: <<[loopnaam]>>
· afsluiten met: END LOOP [loopnaam]
· loop vroegtijdig verlaten: EXIT [loopnaam]
Hoofdstuk 5:  Werken met DBgegevens

1 Select into

· tussen SELECT & FROM clausule
· INTO: vermeldt variabelen die door select statement moeten opgevuld worden

· %type: datatype variabele = datatype kolom op DB
· voorwaarden:

· resultaat select statement: 1 rij
· # variabelen into = # variabelen select
· type boolean ( niet gekend in DB ( niet opnemen in into

INTO [variabele]

· voorbeeld:

declare

l_name emp.ename%type;

l_job emp.job%type;

begin

select ename, job

into l_name, l_job

from emp

where empno='7369';

dbms_output.put_line('Personeelslid: '||l_name||', '||l_job);

end;
2 Expliciete cursors

· 2 soorten cursors
· impliciete cursors:

· volledig door Oracle beheerd

· naam: SQL

· aangemaakt voor elke Select into-statement

· expliciete cursors:

· gedeclareerd in PL/SQL programma
· wanneer er nood is aan rij-per-rij verwerking van resultaattabel van select-statement

· bij gebruik expliciete cursors:

· declaratie

· manipulatie: open, fetch, close
2.1 Declaratie

· in DECLARE
· select-statement: zonder into clausule

CURSOR cur_[naam cursor] IS [select-statement]
2.2 Manipulatie

I Open
OPEN [naam cursor];
· uitvoering select statement
II Fetch

FETCH [naam cursor] INTO [naam variabele(n)];
· eerstvolgende rij uit resultaattabel overbrengen naar variabele
· volgorde moet overeenkomen
III Close

CLOSE [naam cursor];
· gebruikte geheugenruimte terug vrijgeven
2.3 %rowtype

· record variabele = 1 variabele waarin hele gefetchte rij bijgehouden wordt
· zelfde structuur als select statement

rec_[naam record] [naam cursor]%ROWTYPE;

· refereren naar veld binnen record

[naam record].[naam veld]
2.4 Meerdere rijen fetchen

I Basic & while loops
· wanneer er geen rijen meer te fetchen zijn ( laatst gefetchtste rij = actieve rij

a) Basic
· voorbeeld:

declare

  cursor cur_emp is

    select ename, job

    from emp

    order by 1;

  rec_emp cur_emp%rowtype;

begin

  open cur_emp;

  loop
    fetch cur_emp into rec_emp;

    if rec_emp.ename<'M' then

      dbms_output.put_line('Personeelslid: '||rec_emp.ename||', '||rec_emp.job);

    else

      exit;

    end if;

  end loop;

  close cur_emp;

end;

b) While

· voorbeeld:
declare

  cursor cur_emp is

    select ename, job

    from emp

    order by 1;

  rec_emp cur_emp%rowtype;

begin

  open cur_emp;

  fetch cur_emp into rec_emp;

  while rec_emp.ename<'M' loop
    dbms_output.put_line('Personeelslid: '||rec_emp.ename||', '||rec_emp.job);

    fetch cur_emp into rec_emp;

  end loop;

  close cur_emp;

end;

II Cursorattributen

[naam cursor]%FOUND
· boolean

· true: fetch commando vindt nieuwe rij

· false: laatste rij van resultaattabel is verwerkt

· null: fetch commando uitgevoerd, maar nog geen sql commando

· voorbeeld:

declare

  cursor cur_emp is

    select ename, job

    from emp

    order by 1;

  rec_emp cur_emp%rowtype;

begin

  open cur_emp;

  fetch cur_emp into rec_emp;

  while rec_emp.ename<'M' and cur_emp%FOUND loop
    dbms_output.put_line('Personeelslid: '||rec_emp.ename||', '||rec_emp.job);

    fetch cur_emp into rec_emp;

  end loop;

  close cur_emp;

end;

/
[naam cursor]%NOTFOUND
· laatste rij van resultaattabel is verwerkt ( kan geen nieuwe rij gefetched worden
· voorbeeld:

declare

  cursor cur_emp is

    select ename, job

    from emp

    order by 1;

  rec_emp cur_emp%rowtype;

begin

  open cur_emp;

  loop

    fetch cur_emp into rec_emp;

    exit when cur_emp%NOTFOUND;

    if rec_emp.ename<'M' then

      dbms_output.put_line('Personeelslid: '||rec_emp.ename||', '||rec_emp.job);

    end if;

  end loop;

  close cur_emp;

end;
[naam cursor]%ROWCOUNT
· aantal rijen dat reeds gefetched is

[naam cursor]%ISOPEN
· true: cursor is reeds geopend

· null: bij impliciete cursor ( kan niet expliciet via OPEN geopend worden

III For loop

· om alle rijen te doorlopen
· cursor moet niet meer expliciet geopend, gefetched & gesloten worden

FOR [teller] IN [naam cursor] LOOP
[te herhalen code]

END LOOP;

· voorbeeld:

declare 

   cursor cur_emp is 

     select * 

     from emp; 

   l_eerst_gefetcht emp.ename%type; 

   l_laatst_gefetcht emp.ename%type; 

begin 

   for rec_emp in cur_emp loop 

      if cur_emp%rowcount = 1 then 

         l_eerst_gefetcht := rec_emp.ename; 

      end if; 

      l_laatst_gefetcht := rec_emp.ename; 

   end loop; 

   dbms_output.put_line('Eerste personeelslid: '||l_eerst_gefetcht); 

   dbms_output.put_line('Laatste personeelslid: '||l_laatst_gefetcht); 

end; 

2.5 Cursors met parameters

· informatie at runtime in select statement invoeren
· cursor met parameters ( parameters bij uitvoering programma, door actuele waarden vervangen
CURSOR [naam cursor] ([lijst parameters]) IS [select statement]
· declaratie parameter ( in (lijst parameters)
· bij datatype geen precisie ingeven (bv. varchar2)

p_[naam parameter] [datatype] [defaultwaarde]

· voorbeeld:

declare

  cursor cur_job is

    select job

    from emp

    where empno=&empnr;

  l_job emp.job%type;

  cursor cur_emp (p_job varchar2) is

    select *

    from emp

    where job=p_job;

begin

  open cur_job;

  fetch cur_job into l_job;

  close cur_job;

  for rec_emp in cur_emp(l_job) loop

    dbms_output.put_line(rec_emp.empno||': '||rec_emp.ename||', '||rec_emp.hiredate);

  end loop;

end;

/

7369
3 DBgegevens toevoegen, wijzigen, verwijderen

· commando's: INSERT, UPDATE, DELETE ( rechtstreeks in PL/SQL
· specifiek voor PL/SQL: UPDATE, DELETE ( op cursors
3.1 Update & delete gebaseerd op cursors
a) Mogelijkheid 1
· om current rij van cursor te wijzigen/verwijderen
· declaratie cursor: select statement ( aanvullen: 
FOR UPDATE [NOWAIT]
· NOWAIT: rijen door andere gebruiker gelockt ( niet wachten & foutmelding geven

· zorgt ervoor dat gegevens cursor overeenstemmen met gegevens in DB

· update, delete commando ( aanvullen: 
WHERE CURRENT OF [naam cursor];
· voorbeeld:
create table empcopy as select * from emp;

declare

  cursor cur_empcopy is

    select sal, job

    from empcopy

    for update;

begin

  for rec_empcopy in cur_empcopy loop

    if rec_empcopy.sal <=1000 and rec_empcopy.job<>'SALESMAN' then

      update empcopy

      set sal=rec_empcopy.sal*1.02

      where current of cur_empcopy;

    end if;

  end loop;

end;

b) Mogelijkheid 2
· zonder: FOR UPDATE & WHERE CURRENT OF …

· in where: update gelijkstellen aan record_for
· (is trager dan mogelijkheid 1)

· voorbeeld:

declare

  cursor cur_comm is

    select customer.repid, empcopy.ename, count(*) aantal

    from empcopy, customer

    where customer.repid=empcopy.empno

    group by customer.repid, empcopy.ename

    order by 1;

begin

  for rec_comm in cur_comm loop

    if (rec_comm.aantal<>0) then

      update empcopy

      set comm=comm+(100*rec_comm.aantal)

      where empno=rec_comm.repid;

      dbms_output.put_line(rec_comm.ename||' zijn commissie is verhoogd met: '||100*rec_comm.aantal);

    end if;

  end loop;

end;

Hoofdstuk 6:  Exceptions
1 Inleiding
· uitvoeren PL/SQL programma, 2 fasen:

· compilatie fase:

· PL/SQL code vertalen naar pctaal

· ( compilatiefouten

· run fase:

· uitvoeren PL/SQL programma
· ( runtime fouten of exceptions (afh. van actuele waarden)

· exceptions voorzien ( handler schrijven die exception opvangt
2 Afhandeling Oracle-fouten

· geen handler: 

· programma afgebroken

· server message: foutnummer

· exception handler a.h.v. foutnummer
2.1 Exception handlers creëren voor Oracle-fouten
· creëren exception handler:

· declaratie identifier

· koppelen aan Oracle-fout

· codering handler

· bestaan voorgedefinieerde exceptions ( enkel nog coderen
I Declaratie identifier
· in DECLARE gedeelte
· steulelwoord: EXCEPTION
e_[naam exception] EXCEPTION;
II Koppeling aan Oracle-fout
· foutnummer: altijd negatief
uitz: ‘no data found’ ( foutnr. 100
· pragma = compiler directives: geven informatie door aan compiler

· EXCEPTION_INIT pragma: 
· geeft door aan compiler, welke exception aan welke fout gekoppeld is

· in declare gedeelte, na declaratie identifier
· argumenten: identifier, foutnummer

PRAGMA EXCEPTION_INIT(e_[naam exception], [foutnummer]);
III Codering handler

· binnen PL/SQL blok
· sleutelwoord: EXCEPTION ( kan ≠ handlers bevatten
· bij fout: alle handlers, 1 na 1 overlopen

WHEN e_[naam exception] THEN [uit te voeren code];
IV Sqlcode, sqlerrm & others handle

om informatie over opgetreden fout op te halen
a) Sqlcode

· foutnummer actuele Oracle-fout
SQLCODE
b) Sql error mode (sqlerrm)
· tekst die bij bepaalde Oracle-fout hoort
· geen foutnummer meegegeven: tekst huidige fout

SQLERRM([foutnummer])
c) Others handle

· toepassen sqlcode & sqlerrm ( wegschrijven in foutentabel
· handler moet niet gedeclareerd worden

· toevoegen aan EXCEPTION gedeelte ( onvoorziene fouten opvangen
· als laatste handler in EXCEPTION gedeelte
· voorbeeld:
…

exception
  when e_verkeerde_datum THEN dbms_output.put_line('Datum moet van formaat DDMMYY zijn!');

  when others then

    l_code:=sqlcode;

    l_boodschap:=sqlerrm(l_code);

    insert into tblFouten

    values(l_code, l_boodschap);
…
2.2 Voorgedefinieerde exceptions

· enkel nog coderen
· select into levert geen rijen op
· tabel benaderen via onbestaande ingang

NO_DATA_FOUND

· select into ( resultaat > 1 rij

TOO_MANY_ROWS

· delen door 0

ZERO_DEVIDE

· probeert open cursor 2de maal te openen

CURSOR_ALREADY_OPEN

· toevoegen reeds bestaande waarde (on-uniek) aan unieke index 
DUP_VAL_ON_INDEX

· ongeldige operatie op cursor
INVALID_CURSOR

· conversie van string naar nummer is ongeldig
INVALID_NUMBER

· rekenkundige-, numerische-, string-, conversie- of constraint fout
VALUE_ERROR

· incompatibiliteit tussen datatype kolommen & variabelen die ernaar verwijzen
ROWTYPE_MISMATCH

· ongeldige user/paswoord
LOGON DENIED

· poging om DB te benaderen, maar is niet aangelogd
NOT_LOGGOD_ON

· intern probleem
PROGRAM_ERROR

· programmageheugen is vol of corrupt
STORAGE_ERROR

· instance is op abnormale wijze beëindigd
TIMEOUT_ON_RESOURCE
· rollback a.g.v. deadlock
TRANSACTION_BACKED_OUT

· voorbeeld:
declare

  l_ename emp.ename%type;

begin

  select emp.ename

  into l_ename

  from emp

  where ename='Peters';

  dbms_output.put_line(l_ename);

exception
  when no_data_found then dbms_output.put_line('Personeelslid Peters bestaat niet');

  when too_many_rows then dbms_output.put_line('Er zijn meerdere wn met naam Peeters');

end;

3 Programmaverloop bepalen a.h.v. exceptions
· PL/SQL blok afbreken zonder Oracle-fout: 
· declaratie identifier

· codering handler

· activatie handler d.m.v. RAISE: 
RAISE e_[naam exception]

· voorbeeld:
declare

  l_max constant number(1):=5;

  l_aantal_boeken number(1);

  e_max_boek exception;

begin

  if l_aantal_boeken=l_max then

    raise e_max_boek;

--  else  -- registratie boek

  end if;

exception
  when e_max_boek then dbms_output.put_line('U heeft reeds '||l_max||' boeken ontleend. (Dit is het maximum.)');

end;

4 Exception handling binnen geneste blokken

· gebruik maken geneste blokken
· voorbeeld:
<<hoofdblok>>

declare

…

begin

commandoreeks 1

<<subblok>>

declare

…

begin

commandoreeks 2

exception

exceptionhandlers subblok

end subblok;

commandoreeks 3

exception

exceptionhandlers hoofdblok

end hoofdblok;

· fout in subblok:
· exception handlers subblok overlopen

· handler gevonden ( subblok afsluiten, naar commandoreeks 3 (commandoreeks 2)

· geen handler ( exception handlers in hoofdblok overlopen

· exception handlers in hoofdblok overlopen:

· handler gevonden ( hoofdblok afsluiten (commandoreeks 2, 3)

· geen handler ( hoofdblok afsluiten met foutmelding
· blokken waarin verwerking wordt afgebroken voor zoeken naar handler ( worden niet meer uitgevoerd

5 Raise application error
· dbms_output.put_line:
· serveroutput aanzetten

· enkel in sql*plus omgeving zichtbaar

· ( niet geschikt voor sturen van foutmeldingen naar gebruiker

· raise_application_error:
· enkel zinvol tijdens ontwikkelen van programma’s

· procedure speciaal om foutmeldingen naar interface te sturen

· foutnummer: tussen -20 999 & -20 000

· fouttekst: foutboodschap voor gebruiker

· doet: foutboodschap + programma afbreken + rollback (wijzigingen ongedaan gemaakt)
RAISE_APPLICATION_ERROR([foutnummer], [fouttekst]);
· voorbeeld:
declare

  c_te_besteden constant number(3):=100;

  l_totaal number(3):=0;

  cursor cur_empcopy is

    select sal

    from empcopy

    where deptno=20

    for update;

  e_max exception;

begin

  for rec_empcopy in cur_empcopy loop

    if l_totaal<=c_te_besteden then

      update empcopy

      set sal=sal*1.02

      where current of cur_empcopy;

      l_totaal:=l_totaal+(rec_empcopy.sal*0.02);

    else

      raise e_max;

    end if;

  end loop;

exception

  when e_max then raise_application_error(-20000, 'Het te besteden bedrag is overschreden. Loonsverhogingen zullen ongedaan gemaakt worden.');

end;

Hoofdstuk 7:  Functies & procedures
1 Nut van functies & procedures
· lokale: enkel binnen bepaald PL/SQL programma beschikbaar
· stored: over verschilllende programma’s heen beschikbaar
2 Definitie & gebruik van lokale functies & procedures

· declaratie in DECLARE
· NA declaratie van variabelen, constanten, exceptions, …
2.1 Procedures

I Declareren
· syntax gebaseerd op gewoon PL/SQL blok
· steutelwoord: PROCEDURE
· parameterlijst: argumenten van procedure (optioneel)
· EXCEPTION: (optioneel)

· IS: i.p.v. DECLARE
PROCEDURE [naam procedure] ([parameterlijst])
IS
[declaraties]

BEGIN
[uit te voeren code]

EXCEPTION
[exception handlers]

END [naam procedure];
a) Declaratie parameters

p_[naam formele parameter] [mode] [datatype] [DEFAULT [regel]]
· mode: aangeven welke bewerkingen op parameters uitgevoerd mogen worden
· IN: default mode ( enkel lezen

· OUT: om waarde aan toe te kennen ( om resultaten aan hoofdprogramma terug te geven

· IN OUT: uitlezen als toekennen

· datatype: enkel datatype (of %TYPE), geen precisie of schaal
· DEFAULT regel: defaultwaarde voor IN parameter (nooit voor IN OUT of OUT)
b) Formele vs actuele parameters

· formele: zoals ze in definitie van procedure voorkomen
· actuele: actuele waarden die bij aanroepen van procedure meegegeven worden

· voorbeeld:

procedure voeg_emp_toe (p_empno in emp.empno%type, p_ename in emp.ename%type)

is

begin

insert into emp (empno, ename)

values (p_empno, p_ename); ( formele parameters
end;
voeg_pers_toe (‘5909’, ‘Heyns’); ( actuele parameters
II Aanroepen
[naam procedure] ([argumenten]);
· actuele parameters (argumenten) in zelfde volgorde als formele parameters (parameterlijst)
· type actuele parameter afh. mode:
· IN: variabele, constante of expressie

· IN OUT, OUT: variabele

a) Positionele notatie

· parameters met defaultwaarde:
· toch opgenomen in procedure aanroep ( anders problemen met volgorde

· oplossing: parameters met defaultwaarde, als laatste in lijst ( niet altijd juiste oplossing

· op basis van positie

b) Named notation

· op basis van formele naam van parameter
· toekenningsteken: =>
· aanroepen procedure:
[naam procedure] (p_[naam formele parameter] => [actuele parameter]);
· voorbeeld:

procedure invullen_agenda (p_1 in date, p_2 in varchar2 default ‘Vergadering’, p_3 in varchar2 default ‘F913’)

…

invullen_agenda(p_1 => sysdate, p_3 => ‘B5210’);

c) Positionele & named notation samen gebruiken

· eerst argumenten via positionele notatie
· daarna named notation
2.2 Functies declareren

I Declareren

· syntax gebaseerd op gewoon PL/SQL blok

· steutelwoord: FUNCTION
· parameterlijst, exception: (optioneel)

· RETURN: 

· functie levert altijd een waarde op

· enkel datatype, geen precisie of schaal

· in uitvoerbaar deel van functie opnemen + actuele waarde die uit functie resulteert
· telken return commando voorkomen (ook bij exception handlers)

FUNCTION [naam functie] ([parameterlijst]) RETURN [datatype]
IS
[declaraties]

BEGIN
[uit te voeren code]

EXCEPTION
[exception handlers]
 
RETURN ([expressie])
END [naam functie];
II Declaratie parameters
p_[naam formele parameter] [mode] [datatype] [DEFAULT [regel]]

· mode: 

· IN: enkel lezen

· IN OUT & OUT: best niet gebruiken (( opteer dan voor een procedure)

III Functies aanroepen

[naam variabele] := [naam functie] (argumenten);
· voorbeeld:
declare
l_output varchar2(100);

…

function omkeren (p_string in varchar2) return varchar2

is
l_omgekeerd varchar2(100);

begin

for i in reverse 1..length(p_string) loop

l_omgekeerd:=l_omgekeerd||substr(p_string, i, 1);

end loop;

return (l_omgekeerd);

end omkeren;

begin

for rec_pers in cur_pers loop
l_output:=omkeren(rec_pers.naam)
dbms_output.put_line(l_output);

end loop;
end;

3 Stored functions & procedures
· 1 keer voor alle PL/SQL programma’s declareren

· code als object in DB opgeslagen
· procedures, functies als DB opgeslagen ( stored procedures, stored functions
3.1 Voor- & nadelen
a) Voordelen
· vanuit ≠ PL/SQL programma’s, omgevingen opgeroepen worden
· code op 1 plaats opgeslagen: centraal beheren & onderhouden

· performantiewinst:

· gecompileerd opgeslagen (geen compilatie bij uitvoering)

· bij aanroepen in shared pool geladen ( moet bij aanroepen niet telkens weer van schijf gelezen worden

· om beveiliging te verfijnen
b) Nadelen
· hoe meer objecten opgeslagen ( hoe meer schijfruimte
· voor optimale performantiewinst ( werkgeheugen DBomgeving uitbreiden
3.2 Creatie & gebruik van DB functies & procedures

I Creatie

CREATE [OR REPLACE] FUNCTION [ … functie declaratie …];
CREATE [OR REPLACE] PROCEDURE [ … procedure declaratie …];
· syntax zelfde als voor locale functie/procedure
· sleutelwoord: CREATE [OR REPLACE]

· REPLACE: 

· oudere versies overschreven
· toegekende rechten blijven behouden

· zoniet (als er oudere versies bestaan) ( nieuwe kan niet aangemaakt worden

· broncode in data-dictionary opgeslagen (ongeacht fouten): USER_SOURCE
· compilatie slaagt: gecompileerde code in systemtabellen
· compilatie mislukt:

· foutinformatie naar data-dictionary: USER_ERRORS
· foutmelding

SHOW ERRORS [PROCEDURE | FUNCTION [naam procedure] | [naam functie] ]
· compilatiefouten uitlijsten

· geen parameters ( meest recente compilatiefout

· geen ‘;’

II Verwijderen
DROP PROCEDURE [naam procedure];
DROP FUNCTION [naam functie];
· ook alle toegekende rechten worden verwijderd

III Gebruik
(via dual)
select omkeren('kat')

from dual;
· uitvoeren zonder apart PL/SQL blok te schrijven, waarin functie/procedure opgeroepen wordt
a) Informatie
DESCRIBE [naam functie | naam procedure]
· oproepen a.h.v. naam
· nodig om functie/procedure correct aan te roepen:
· type (functie/procedure)
· naam

· argumentenlijst

· bij functie: datatype return

· ook te vinden in DD: USER_OBJECTS
b) Procedures zonder OUT
EXECUTE [naam procedure] ([IN parameters]);
· voorbeeld:

function nieuwe_verkoper(p_naam in varchar2, p_voornaam in varchar2, p_salaris in number)

…

end nieuwe_verkoper

execute nieuwe_verkoper(‘Verbiest’, ‘Vincent’, ‘200’);
c) Procedures met OUT
EXECUTE [naam procedure] ([in parameters], :[variabelen out]);
VARIABLE v_[naam variabele] [datatype]
· variabelen declareren: 
· voor stockeren waarden van OUT parameters
· enkel datatype: number, char, varchar2

· referereren door: :[naam variabele]
PRINT v_[naam variabele]

· inhoud variabele opvragen

· voorbeeld:

function verhoog_sal(p_persno in varchar2, p_procent in number, p_message out varchar2)

…

end verhoog_sal
variable v_melding varchar2(100);

execute verhoog_sal(‘5903’, ‘3’, :v_melding);
print v_melding;
d) Functies

EXECUTE :[naam variabele] := [naam functie];
VARIABLE v_[naam variabele] [datatype]

· variabelen declaren voor stockeren return waarde
· voorbeeld:
variable v_omgekeerd varchar2(100);

execute :v_omgekeerd := omkeren(‘patat’);

3.3 Stored functions in SQL statements
· voorwaarden voor functie:
· parameterlijst met enkel IN parameters

· ingeven actuele parameters: named notation
· datatype parameters + return: boolean, record, PL/SQL tabel
· geen COMMIT/ROLLBACK uitvoeren

· inhoud tabellen niet wijzigen

· anders foutmelding: Function … does not guarantee not to update database.
3.4 Data dictionary views

· via select-statements opvragen
a) User source

USER_SOURCE

· broncodes van functie/procedure van aangelogde gebruiker

· kolommen: name, object_type, line, text

b) User errors

USER_ERRORS

· compilatiefouten

· verwijderd wanneer: 

· procedure/functie succesvol gecompileerd is

· procedure/functie verwijderd wordt

c) User obects
USER_OBJECTS

· overzicht alle objecten door aangelogde gebruiker aangemaakt
· PL/SQL objecten, tabellen, views, …

· kolom: status (valid / invalid) ( valid: kan refereren naar object
d) User independencies
USER_INDEPENDENCIES

· afhankelijkheden registreren
· procedures, functies, packages, views ( als afhankelijk en/of gerefereerd object
· tabellen, sequences ( enkel als gerefeerde objecten

· per afhankelijk object: naam, type, eigenaar (van gerefereerd object)
· wijziging structuur object ( afhankelijke objecten op invalid
· gecompileerde code niet meer geldig

· opnieuw compileren:

ALTER PROCEDURE | FUNCTION [naam procedure | naam functie] COMPILE;
· alle invalid objecten van bepaald schema hercompileren
BEGIN
DBMS_UTILITY.COMPILE_SCHEMA ([naam schema]);
END;
· broncode wijzigen:
· object opnieuw creëren: CREATE OR REPLACE …
· object verwijderen: DROP … + aanmaken: CREATE …
Hoofdstuk 8:  Database triggers
1 Inleiding
· = PL/SQL programmadat aan tabel is verbonden
· als inhoud tabel wijzigt (door insert, update of delete) ( trigger wordt uitgevoerd

· doeleinden:

a) Gegevensauditing

· commando AUDIT: bijhouden welke gebruikers, welke bewerkingen op welke objecten uitvoeren
· trigger: wijzigingen zelf in tabel bijhouden
b) Implementatie dynamische constraints

· = beperkingen aan overgang van 1 status van DB, naar andere (nieuwe status)
· kan niet met commando CREATE vastgelegd worden

· implementeren a.h.v. trigger
c) Implementatie extra beveiliging

· serverniveau: toegang tot tabellen a.h.v. rechten
· triggers: beveiligen waarvoor geen overeenkomstige rechten zijn
d) Automatische uitvoering gerelateerde acties

· wijzigen gegevens in tabel ( bewerkingen in andere tabellen, andere acties vereisen
2 DB triggers creëren

2.1 Kenmerken

I Triggering event

· bij creatie trigger

· actie die aanleiding geef tot uitvoeren trigger code

· eventen: INSERT, UPDATE, DELETE
· eventen combineren: OR
II Trigger timing

· bij creatie trigger

· steutelwoorden: BEFORE, AFTER
· code vóór of nà triggering event uitvoeren?

III Trigger type

· statement trigger:

· gaat 1 keer af bij event
· # bewerkte rijen ( geen invloed op # maal dat trigger-code uitgevoerd wordt

· row trigger:

· trigger-code wordt voor elke bewerkte rij uitgevoerd
· sleutelwoord: FOR EACH ROW
IV Trigger code/actie
· wat moet gebeuren als trigger afgaat?

· functies/procedures oproepen

2.2 Syntax

CREATE [OR REPLACE] TRIGGER [naam trigger]
BEFORE | AFTER
INSERT [OR] | DELETE [OR] | UPDATE [OF [kolomnamen]]

ON [tabel naam]

[FOR EACH ROW [WHEN [voorwaarde]]]

[DECLARE]

BEGIN
[EXCEPTION]

END [naam trigger];
· voorbeeld:
select to_char(sysdate,'D') from dual; --kijken welk dag het is

create or replace trigger controle_weekdag

  before update of sal on emp

begin
  if to_char(sysdate,'D') IN ('6', '7') then

    raise_application_error(-20000, 'Salaris kan niet in weekend gewijzigd worden.');

  end if;

end controle_weekdag;

update emp

set sal=sal*1;

--> .ORA-20000: Salaris kan niet in weekend gewijzigd worden.
3 Beperkingen voor triggers

3.1 Algemene beperking

· mogen geen COMMIT/ROLLBACK bevatten
· geen procedures/functies aanroepen met COMMIT/ROLLBACK
3.2 Beperking voor rij-triggers

FOR EACH ROW WHEN OLD.[kolomnaam] | NEW.[kolomnaam]

:OLD.[kolomnaam] | :NEW.[kolomnaam]
· mogelijk om voor elke rij oorspronkelijke & nieuwe waarden te lezen

· steutelwoorden: OLD, NEW
· in FOR EACH ROW WHEN: geen ‘:’

· trigger timing:

· BEFORE: zelf nieuwe waarde bepalen

· AFTER: enkel nieuwe waarde nog lezen

· 2 beperkingen:
· mutating table

· constraining table

I Beperking van mutating table
· = wanneer tabel bewerkt wordt
· mutating table:
· DELETE: from …

· INSERT: into …

· UPDATE: update …

· 1 DML-commando kan leiden tot ≠ resultaten (afh. van volgorde records)( voor ≠ mutating tables
· voorbeeld: salaris verhoging ( volgorde records

· row triggers mogen niet lezen in mutating table
· fout: table … is mutating, trigger/function may nog see it.
II Beperking van constraining table

· = andere tabel waaruit rij-trigger van ene tabel leest
· voorbeeld: code gemeentes verandert automatisch in personeel

· row triggers mogen nooit PK, FK of UK wijzigingen in constraining table

· fout: table … is constraining, trigger may not modify it.

Hoofdstuk 9:  Oefeningen – gedoe
1 Functies
· te vervangen kan maar 1 char zijn, of opeenvolgende chars

REPLACE ([identifier], [te vervangen], [vervangen door]);

· te vervangen mogen chars zijn, die elkaar niet opvolgen

· elke te vervangen, respectievelijk, elke vervangen door

TRANSLATE ([identifier], [te vervangen], [vervangen door]);

· rest berekenen

MOD([getal], [deler])

· lengte string
LENGTH([variabele])

· substring

SUBSTR([string], [karakter], [aantal karakters])

2 Sql-gedoe
· tabel aanmaken als kopie
CREATE TABLE [tabelnaam] AS SELECT [kolomnaam] FROM [tabelnaam];

· nieuwe tabel aanmaken

CREATE TABLE [tabelnaam]

([kolomnaam] [datatype], …);
· kolommen van tabel bekijken

DESC[RIBE] [tabelnaam]
· tabel updaten

· alle rijen worden geupdated (tenzij where statement ingevuld)

UPDATE [tabelnaam]

SET [kolomnaam] = …;
· afzonderlijke waarden invoegen in bestaande tabel
INSERT INTO [tabelnaam]

VALUES ([waarden]);

· rijen deleten uit tabel
· zonder where ( hele tabel gedeleted

DELETE FROM [tabelnaam]

WHERE [voorwaarde];
· commit:
· alle wijzigingen vastgelegd

· savepoints worden verwijderd

· alles vóór commit kan nooit meer gewijzigd worden

COMMIT;
· savepoint:

· identificeert punt in transactie tot waar gerollbackt kan worden
SAVEPOINT [naam savepoint];
· rollback:

· annuleert wijzigingen

· tot aan laatste commit of gedefinieerd savepoint

ROLLBACK [TO [naam savepoint]];
· select

SELECT [ALL | DISTINCT] [kolomnamen]
FROM [tabelnaam]
WHERE [opties]
GROUP BY [kolomnamen]

HAVING [voorwaarden]

ORDER BY [kolomnaam] [ASC | DESC]

